
;****************************************************************************
;*Πρόγραμμα μέτρησης θερμοκρασίας με το LM335 και απεικόνιση σε ενδείκτες   *
;*επτά τμημάτων (SSD).                                                      *
;*                                                                          *
;*     Όνομα αρχείου : tempSSD.asm                                          *
;*        Δημιουργία : Οκτώβριος 2004                                       *
;*Τελευταία ενημέρωση: Μάρτιος 2005                                         *
;*        Συγγραφέας : Α.Μαρμαρινός                                         *  
;****************************************************************************

#INCLUDE "p16f877.inc" ; Συμπεριέλαβε το αρχείο δηλώσεων του επεξεργαστή
;****************************************************************************
;*CONFIG: Ρύθμιση διαφόρων παραμέτρων του μικροελεγκτή                      *
;*  CP:     Code Protect - Προστασία κώδικα                                 *
;*  DEBUG:  In Circuit Debugger Mode - Κατάσταση εκσφαλμάτωσης              *
;*  WRT:    FLASH Program Memory Write Enable - Επίτρεψη εγγραφής στη μνήμη *
;*  CPD:    EE Memory Code Protect - Προστασία κώδικα στη μνήμη EEPROM      *
;*  LVP:    Low Voltage Programing - Προγραμματισμός με χαμηλή τάση         *
;*  BODEN:  Brown out Detect Enable                                         * 
;*  PWRTE:  Power up timer Enable                                           *
;*  WDT:    Watchdog timer                                                  *
;*  RC_OSC: RC Oscilator - Ταλαντωτής RC                                    *
;*  HS_OSC: High speed crystal Oscilator - Ταλαντωτής υψηλής συχνότητας     *
;*  XT_OSC: Crystal Oscilator - Κρυσταλλικός ταλαντωτής                     *
;*  LP_OSC: Low Power Crystal Oscilator - Ταλαντωτής χαμηλής ισχύος         *
;****************************************************************************

__CONFIG h'3DB9';_CP_OFF & _DEBUG_OFF & _WRT_ENABLE_OFF & _CPD_OFF 
;& _LVP_ON & _BODEN_OFF & _PWRTE_OFF & _WDT_OFF & _XT_OSC

;*****************************************************************************
; Δηλώσεις Καταχωρητών - Μεταβλητών του προγράμματος
;*****************************************************************************
;
;Καταχωρητές για τη μετατροπή από δυαδικό σε BCD
BCD10 equ 0x020 ; Για την αποθήκευση των εκατοντάδων
BCD10 equ 0x021 ; Για την αποθήκευση των δεκάδων
BCD1 equ 0x022 ; Για την αποθήκευση των μονάδων
;Διάφοροι καταχωρητές
ADRES equ 0x023 ; Για την αποθήκευση του αποτελέσματος της μετατροπής A/D
DELAY equ 0x024 ; Χρήση στη ρουτίνα DELAY500 
X_DELAY equ 0x025 ; Χρήση στη ρουτίνα X_DELAY500

;*****************************************************************************
; Έναρξη προγράμματος
;*****************************************************************************

ORG 0x0000 ; Θέση του RESET vector 
RESET goto START

;*****************************************************************************
; Ρουτίνα εξυπηρέτησης διακοπών
;*****************************************************************************

ORG 0x0004 ; Θέση του Interrupt vector 
INTERRUPT goto    INTERRUPT

;*****************************************************************************
; Αρχικοποίηση καταχωρητών του επεξεργαστή. Aρχικοποίηση μονάδας A/D
;*****************************************************************************
START ; POWER_ON Reset (Αρχή του προγράμματος) 

BANKSEL PORTD ; Επιλογή τμήματος μνήμης 0
clrf PORTD ; Οι θύρες C,D σε κατάσταση Low.
clrf PORTC
movlw B'01000001' ; Ρολόι μετατροπής:Fosc/8, Ενεργοποίηση A/D,...
movwf ADCON0 ; ...δειγματοληψία από το κανάλι 0
BANKSEL TRISD ; Επιλογή τμήματος μνήμης 1 (εκεί βρίσκεται ο TRISD)
clrf TRISD ; Οι θύρες C,D καθορίζονται ως έξοδοι
clrf TRISC 
movlw B'00001111' ; Αριστερή στοιχιση, 1 αναλογική είσοδος...
movwf ADCON1 ; ...RA3(=5V) και RA2(=2.45V) τάσεις αναφοράς
BANKSEL PORTD ; Επιλογή τμήματος μνήμης 0

;*********************************************************************************
;Διαδικασία A/D μετατροπής επεξεργασίας και απεικόνισης των μετρήσεων
;*********************************************************************************



Loop bsf ADCON0,GO ; Έναρξη μετατροπής A/D στο κανάλι 0
Wait btfss PIR1,ADIF ; Έλεγχος της σημαίας ολοκλήρωσης της μετατροπής (ADIF)

goto Wait ; Αναμονή για την ολοκλήρωση της μετατροπής
movf ADRESH,W ; Διάβασε το αποτέλεσμα της μετατροπής...
movwf ADRES ; ...και αποθήκευσέ το στον καταχωρητή ADRES
bcf STATUS, C ; ΜΗΔΕΝΙΣΜΌΣ C
movlw h'1C' ;-|
subwf ADRES , F ; | ΑΦΑΙΡΕΣΗ (ADRES-1C) (δηλ d'28')
btfss STATUS, C ;-|   (2.73-2.45)*(255/2.55)=28
goto NEGATIVE ; Αν ο αριθμός είναι αρνητικός πήγαινε στην κατάλληλη ρουτίνα

;
movf ADRES,W ; Μεταφορά του ADRES στον W
call bin2bcd8 ; Κλήση της υπορουτίνας μετατροπής από δυαδικό σε BCD

; Απεικόνιση δεκάδων
movf BCD10, W ; Ανάγνωση της BCD τιμής των δεκάδων
call Decode ; Κλήση υπορουτίνας αποκωδικοποίησης
movwf PORTC ; Απεικονιση δεκάδων

; Απεικόνιση μονάδων
DISP movf BCD1, W ; Ανάγνωση της BCD τιμής των μονάδων

call Decode ; Κλήση υπορουτίνας αποκωδικοποίησης
movwf PORTD ; Απεικόνιση μονάδων

movlw h'FF'
call X_DELAY500 ; Καθυστέρηση μέχρι την επόμενη μέτρηση
call X_DELAY500 ; Συνολικά (255*0.5)+3*(165*0.5)=375 msec
call X_DELAY500
call X_DELAY500

bcf PIR1,ADIF ; Εκκαθάριση της σημαίας ολοκλήρωσης της μετατροπής 
goto Loop ; Επανάληψη της διαδικασίας

NEGATIVE
movlw B'11111110' ; Απεικόνιση του πλην '-'
movwf PORTC
movf ADRES, W ; Μεταφορά του ADRES στον W
sublw H'FF' ; Αντιστροφή του αριθμού:Αφαίρεση FF-W -> W 
addlw H'1' ;                         πρόσθεση W+1 -> W
movwf ADRES ; μεταφορά στον ADRES

call bin2bcd8 ; Κλήση της υπορουτίνας μετατροπής από δυαδικό σε BCD
goto DISP ; Η συνέχεια είναι όπως και στους θετικούς

;*********************************************************************************
; Decode
; Ρουτίνα αποκωδικοποίησης. Ενεργοποιεί τα αντίστοιχα bit ώστε να εμφανιστεί
; στον ενδείκτη επτά τμημάτων ο αντίστοιχος κάθε φορά αριθμός.
;*********************************************************************************
Decode addwf PCL,F

retlw B'10000001' ;Αποκωδικοποίηση του αριθμού 0 
retlw B'11001111' ;Αποκωδικοποίηση του αριθμού 1 
retlw B'10010010' ;Αποκωδικοποίηση του αριθμού 2 
retlw B'10000110' ;Αποκωδικοποίηση του αριθμού 3 
retlw B'11001100' ;Αποκωδικοποίηση του αριθμού 4 
retlw B'10100100' ;Αποκωδικοποίηση του αριθμού 5 
retlw B'11100000' ;Αποκωδικοποίηση του αριθμού 6
retlw B'10001111' ;Αποκωδικοποίηση του αριθμού 7 
retlw B'10000000' ;Αποκωδικοποίηση του αριθμού 8 
retlw B'10001100' ;Αποκωδικοποίηση του αριθμού 9 

;*****************************************************************************
; bin2bcd8
;
; Ρουτίνα μετατροπής ενός 8-bit αριθμού σε 3-ψήφιο BCD
; Μετατρέπει το περιεχόμενο του W και αποθηκεύει τις εκατοντάδες στον
; καταχωρητή BCD100, τις δεκάδες στον BCD10 και τις μονάδες στον BCD1
; Π.χ. ο αριθμός h'C2'=b'11000010'=d'194' μετατρέπεται σε (01,09,04)
; δηλ. 1 εκατοντάδα, 9 δεκάδες, 4 μονάδες
;
;*******************************************************************************
;
bin2bcd8



clrf  BCD100 ; Μηδένισε τους καταχωρητές που θα γραφεί το αποτέλεσμα
clrf BCD10 ; 
clrf BCD1
bcf STATUS,C ; Μηδένισε τη σημαία κρατουμένου

EKATONTADES ; Υπολογισμός εκατοντάδων
addlw -.100 ; Αφαίρεσε μια εκατοντάδα
btfss STATUS,C ; Αρνητικό αποτέλεσμα;
goto DEKADES ; Αν ναι, τέλειωσαν οι εκατοντάδες - υπολόγισε τις δεκάδες 
incf BCD100, F ; Αν όχι, αύξησε το μετρητή εκατοντάδων...
goto EKATONTADES ; ...και επανέλαβε

DEKADES ; Υπολογισμός δεκάδων 
addlw .100 ; Επανέφερε τον αριθμό
bcf STATUS,C ; Μηδένισε τη σημαία κρατουμένου

again
addlw   -.10 ; Αφαίρεσε μια δεκάδα
btfss   STATUS,C ; Αρνητικό αποτέλεσμα;
goto    MONADES ; Αν ναι, τέλειωσαν οι δεκάδες και μένουν οι μονάδες
incf    BCD10, F ; Αν όχι, αύξησε το μετρητή δεκάδων...
goto    again ; ... και επανέλαβε

MONADES
addlw   .10 ; Επανέφερε τον αριθμό
movwf   BCD1 ; Έχουν μείνει μόνο οι μονάδες, οπότε αποθήκευσέ τις 
return

;****************************************************************************
; Ρουτίνα χρονικής καθυστέρησης
;*****************************************************************************
; Η διάρκεια ενός κύκλου είναι 1 μsec για κρύσταλλο 4MHz
; Fosc = 4MHz
; Cycle_time= 1/(Fosc / 4) = 1/[(4*10^6)/ 4] = 1μsec
;*****************************************************************************
; Delay_time = ((DELAY_value * 3) + 4) * Cycle_time
; DELAY_value = (Delay_time - (4 * Cycle_time)) / (3 * Cycle_time)
;
; π.χ. (για κρύσταλλο 4MHz)
; Delay_time = ((32 * 3) + 4) * 1μsec
; = 100μsec
; DELAY_value = (500μsec - 4) / 3
; = 165.33
; = 165
;*****************************************************************************
; Ρουτίνα   DELAY500 : Καθυστέρηση 500 μsec
; Ρουτίνα Χ_DELAY500 : Καθυστέρηση Χ*500 μsec, όπου Χ είναι το περιεχόμενο του W
;
;*****************************************************************************
DELAY500 movlw D'165' ; +1 1 cycle

movwf DELAY ; +2 1 cycle
DELAY500_LOOP decfsz DELAY, F ; step 1 1 cycle

goto DELAY500_LOOP ; step 2 2 cycles
DELAY500_END return ; +3 2 cycles
;
;
X_DELAY500 movwf X_DELAY ; +1 1 cycle
X_DELAY500_LOOP call DELAY500 ; step1 wait 500μsec

decfsz X_DELAY, F ; step2 1 cycle
goto X_DELAY500_LOOP ; step3 2 cycles

X_DELAY500_END return ; +2 2 cycles
;*****************************************************************************

END ; Τέλος του προγράμματος


